Solving differential equations with MAPLE

Functions and graphic representation

The derivation of the functions

Initialization of the solving ODE package

```
> restart:
```
clears the memory of all previously saved values and variables

```
> with(DEtools):
```
load the differential equations package

```
> with(plots):
```
load the graphical package

```
Warning, the name changecoords has been redefined
```

Define and solve a first order differential equation

Let consider the differential equation $\frac{d}{dx}\mathrm{y}(x) = k\,\mathrm{y}(x)$ where k is a real coeficient. The differential equation can be introduse in MAPLE as follows:

```
> diff_eq1:=diff(y(x),x) = k*y(x);
```

$$diff\_eq1 := \frac{d}{dx}\mathrm{y}(x) = k\,\mathrm{y}(x)$$

To obtain the general solution of the equation use **dsolve** command

```
> dsolve(diff_eq1,y(x));
```

$$\mathrm{y}(x) = \_C1\,\mathbf{e}^{(k\,x)}$$

The general solution is seen as an expresion. Notice that the undetermined constant is called *_C1*
How can we manipulate this expresion?
We can use the function definition command:

```
> sol:=(x,k,c)->c*exp(k*x);
```

$$sol := (x, k, c) \rightarrow c\,\mathbf{e}^{(k\,x)}$$

If the expresion of the solution is too complicated we can use the command **rhs** (*right hand side*) and **unapply** in order to obtain the solution as a function

```
> right_hand_expr:=rhs(dsolve(diff_eq1,y(x)));
```

$$right\_hand\_expr := \_C1\,\mathbf{e}^{(k\,x)}$$

Using the **unapply** command we transform the expresion sol1 into a function specifying the variables:

```
> sol1:=unapply(right_hand_expr,x,k,_C1);
```

$$sol1 := (x, k, \_C1) \rightarrow \_C1\,\mathbf{e}^{(k\,x)}$$

and we get the same result.

The graphics of ODE solutions

Let suppose that $k := 2$. Then the corresponding general solution is:

```
> y:=(x,c)->sol(x,2,c);
```

$$y := (x, c) \rightarrow \mathrm{sol}(x, 2, c)$$

To draw the solutions curves you just assign some values for the constant c. For example take c:=1 c:=2 and c:=-1

```
> plot([y(x,1),y(x,2),y(x,-1)],x=-2..2);
```



```
> animate( y(x,c),x=-2..2,c=-2..2,frames=50);
```



If you want to obtain the solutions with some specified colors use the command:

```
> plot([y(x,1),y(x,2),y(x,-1)],x=-2..2,color=[black,red,blue]);
```



Also you specify the window of the graphic:

```
> plot([y(x,1),y(x,2),y(x,-1)],x=-2..2,y=-
10..10,color=[black,red,blue]);
```



Using this way of manipulation for the solution you can see also how the solutiond depends on the k parameter. Let us consider c:=1 and assign some values for th parameter k.

```
> y1:=(x,k)->sol(x,k,1);
```

$$y1 := (x, k) \rightarrow \mathrm{sol}(x, k, 1)$$

```
> plot([y1(x,0.2),y(x,0.5),y(x,-1)],x=-2..2,y=-10..10,color=[black,red,blue]);
```



Solving an IVP

Suppose that we want to solve the IVP $\frac{d}{dx}\mathrm{y}(x) = k\,\mathrm{y}(x)$ with the initial condition $\mathrm{y}(0) = 1$

```
> restart:with(DEtools):
> diff_eq:=diff(y(x),x) = k*y(x);
```

$$diff\_eq := \frac{d}{dx}\mathrm{y}(x) = k\,\mathrm{y}(x)$$

```
> in_cond:=y(0)=1;
```

$$in\_cond := \mathrm{y}(0) = 1$$

```
> dsolve({diff_eq,in_cond},y(x));
```

$$\mathrm{y}(x) = \mathbf{e}^{(k\,x)}$$

Let consider the case $k = 2$

```
> k:=2;
```

$$k := 2$$

```
> sol:=dsolve({diff_eq,in_cond},y(x));
```

$$sol := y(x) = e^{(2x)}$$

```
> yy:=x->rhs(sol);
```

$$yy := x \rightarrow rhs(sol)$$

```
> plot(yy(x),x=-1..1,y=0..4);
```



You can obtain the graph the IVP directly using the command DEplot:

```
> DEplot(diff_eq,y(x),x=-1..1,[[in_cond]]);
```



In this graph is also represented the direction field of the equation. If you want the graphs of the solutions for different initial condition ($y(0)=1$, $y(0)=1.5$, $y(0)=2$) you can use the same command and specify the list of initial conditions:

```
> DEplot(diff_eq,y(x),x=-1..1,[[y(0)=1],[y(0)=1.5],[y(0)=2]]);
```



>

First Order Solvable Differential Equations

Separable Differential Equations

$$\frac{d}{dx}\,\mathrm{y}(x) = \mathrm{f}(x)\,\mathrm{g}(\mathrm{y}(x))$$

```
> restart:with(DEtools):
> sep_eq:=diff(y(x),x)=f(x)*g(y(x));
```

$$sep\_eq := \frac{d}{dx}\,\mathrm{y}(x) = \mathrm{f}(x)\,\mathrm{g}(\mathrm{y}(x))$$

```
> dsolve(sep_eq,y(x));
```

$$\int \mathrm{f}(x)\,dx - \int^{\mathrm{y}(x)} \frac{1}{\mathrm{g}(\_a)}\,d\_a + \_C1 = 0$$

Examples:

a) $y' = 2\,x\,(1+y^2)$

b) $(x^2-1)\,y' + 2\,x\,y^2 = 0$

c) $y' = e^{(x+y)}$

```
> eq_a:=diff(y(x),x)=2*x*(1+(y(x))^2);
```

$$eq\_a := \frac{d}{dx}\,\mathrm{y}(x) = 2\,x\,(1+\mathrm{y}(x)^2)$$

```
> dsolve(eq_a,y(x));
```

$$\mathrm{y}(x) = \tan(x^2 + 2\,\_C1)$$

```
> eq_b:=(x^2-1)*diff(y(x),x)+2*x*(y(x))^2=0;
```

$$eq\_b := (x^2-1)\left(\frac{d}{dx}\,\mathrm{y}(x)\right) + 2\,x\,\mathrm{y}(x)^2 = 0$$

```
> dsolve(eq_b,y(x));
```

$$y(x) = \frac{1}{\ln(x-1) + \ln(x+1) + \_C1}$$

```
> eq_c:=diff(y(x),x)=exp(x+y(x));
```

$$eq\_c := \frac{d}{dx}\,y(x) = \mathbf{e}^{(x+y(x))}$$

```
> dsolve(eq_c,y(x));
```

$$y(x) = \ln\left(-\frac{1}{\mathbf{e}^{x} + \_C1}\right)$$

Homogeneous (in the Euler sense) Differential Equations

$$\frac{d}{dx}\,y(x) = F\left(\frac{y(x)}{x}\right)$$

```
> hom_eq:=diff(y(x),x)=F(y(x)/x);
```

$$hom\_eq := \frac{d}{dx}\,y(x) = F\left(\frac{y(x)}{x}\right)$$

```
> dsolve(hom_eq,y(x));
```

$$y(x) = \mathrm{RootOf}\left(-\int^{\_Z} \frac{1}{F(\_a) - \_a}\, d\_a + \ln(x) + \_C1\right) x$$

Examples:

a) $2\,x^2\;\;y' = x^2 + y^2$

b) $x\;\;y' = \sqrt{x^2 - y^2} + y$

c) $2\,x^3\;\;y' = y^3 + x^2\,y$

```
> eq_a:=2*x^2*diff(y(x),x)=x^2+y(x)^2;
```

$$eq\_a := 2\left(\frac{d}{dx}\,y(x)\right)x^2 = x^2 + y(x)^2$$

```
> dsolve(eq_a,y(x));
```

$$y(x) = \frac{x\,(-2 + \ln(x) + \_C1)}{\ln(x) + \_C1}$$

```
> eq_b:=x*diff(y(x),x)=sqrt(x^2-y(x)^2)+y(x);
```

$$eq\_b := x\left(\frac{d}{dx}\,y(x)\right) = \sqrt{x^2 - y(x)^2} + y(x)$$

```
> dsolve(eq_b,y(x));
```

$$-\arctan\left(\frac{y(x)}{\sqrt{x^2 - y(x)^2}}\right) + \ln(x) - \_C1 = 0$$

```
> eq_c:=2*x^3*diff(y(x),x)=y(x)^3+x^2*y(x);
```

$$eq\_c := 2\,x^3 \left(\frac{d}{dx}\,y(x)\right) = y(x)^3 + y(x)\,x^2$$

```
> dsolve(eq_c,y(x));
```

$$y(x) = \frac{x}{\sqrt{1 + \_C1\,x}},\; y(x) = -\frac{x}{\sqrt{1 + \_C1\,x}}$$

First Order Linear Differential Equations

$$\left(\frac{d}{dx}\,y(x)\right) + P(x)\,y(x) = Q(x)$$

```
> lin_eq:=diff(y(x),x)+P(x)*y(x)=Q(x);
```

$$lin\_eq := \left(\frac{d}{dx}\,y(x)\right) + P(x)\,y(x) = Q(x)$$

```
> dsolve(lin_eq,y(x));
```

$$y(x) = \left(\int Q(x)\,\mathbf{e}^{\left(\int P(x)\,dx\right)}\,dx + \_C1\right)\mathbf{e}^{\left(\int -P(x)\,dx\right)}$$

Examples:

a) $y' + y\tan(x) = \frac{1}{\cos(x)}$

b) $y' + \frac{y(x)}{x} = 3\,x$

c) $x\,y' + y = \mathbf{e}^x$

```
> eq_a:=diff(y(x),x)+tan(x)*y(x)=1/cos(x);
```

$$eq\_a := \left(\frac{d}{dx}\,y(x)\right) + \tan(x)\,y(x) = \frac{1}{\cos(x)}$$

```
> dsolve(eq_a,y(x));
```

$$y(x) = \cos(x)\tan(x) + \cos(x)\,\_C1$$

```
> eq_b:=diff(y(x),x)+1/x*y(x)=3*x;
```

$$eq\_b := \left(\frac{d}{dx}\,y(x)\right) + \frac{y(x)}{x} = 3\,x$$

```
> dsolve(eq_b,y(x));
```

$$y(x) = \frac{x^3 + \_C1}{x}$$

```
> eq_c:=x*diff(y(x),x)+y(x)=exp(x);
```

$$eq\_c := x\left(\frac{d}{dx}\,y(x)\right) + y(x) = \mathbf{e}^x$$

```
> dsolve(eq_c,y(x));
```

$$y(x) = \frac{\mathbf{e}^x + \_C1}{x}$$

\>

Solving a second order ODE

```
> restart:
> with(DEtools):
> with(plots):
Warning, the name changecoords has been redefined
```

Consider the linear differential equation with the constant coeficients y '' + 3 y ' + 2 y = $1+x^2$

```
> deq1:=diff(y(x),x$2)+3*diff(y(x),x)+2*y(x)=1+x^2;
```

$$deq1 := \left(\frac{d^2}{dx^2}\,\mathrm{y}(x)\right) + 3\left(\frac{d}{dx}\,\mathrm{y}(x)\right) + 2\,\mathrm{y}(x) = 1 + x^2$$

To obtain the general solution we use the dsolve command

```
> dsolve(deq1,y(x));
```

$$\mathrm{y}(x) = \frac{9}{4} + \frac{x^2}{2} - \frac{3\,x}{2} - \mathbf{e}^{(-2\,x)}\,\_C1 + \mathbf{e}^{(-x)}\,\_C2$$

If we want to study the solution we can use the same technique as in the previous section in order to draw the solution graph.

```
> sol:=dsolve(deq1,y(x));
```

$$sol := \mathrm{y}(x) = \frac{9}{4} + \frac{x^2}{2} - \frac{3\,x}{2} - \mathbf{e}^{(-2\,x)}\,\_C1 + \mathbf{e}^{(-x)}\,\_C2$$

```
> right_hand:=rhs(sol);
```

$$right\_hand := \frac{9}{4} + \frac{x^2}{2} - \frac{3\,x}{2} - \mathbf{e}^{(-2\,x)}\,\_C1 + \mathbf{e}^{(-x)}\,\_C2$$

```
> y_sol:=unapply(right_hand,x,_C1,_C2);
```

$$y\_sol := (x, \_C1, \_C2) \rightarrow \frac{9}{4} + \frac{1}{2}\,x^2 - \frac{3}{2}\,x - \mathbf{e}^{(-2\,x)}\,\_C1 + \mathbf{e}^{(-x)}\,\_C2$$

Now we are able to one ore more than one solution graphs using the **plot** command.

```
> plot([y_sol(x,0,0),y_sol(x,0,1),y_sol(x,1,0)],x=-2..2,y=-10..10);
```



In the case of initial value problem we have two initial conditions, for example lets take $\mathrm{y}(0) = 1$ and y ' (0) = 0.

```
> in_cond:=y(0)=1,D(y)(0)=0;
```

$$in\_cond := y(0) = 1, D(y)(0) = 0$$

To obtain the corresponding solution we use **dsolve** command in the following form:

```
> dsolve({deq1,in_cond},y(x));
```

$$y(x) = \frac{9}{4} + \frac{x^2}{2} - \frac{3x}{2} - \frac{1}{4} e^{(-2x)} - e^{(-x)}$$

Now we can use the previous technique (**rhs** and **unapply** comands) to construct the solution as a function and after that to represent its graph or we can obtain this graph directly using **DEplot** command.

```
> DEplot(deq1,y(x),x=-2..2,y=-10..10,[[in_cond]]);
```



If we need to draw more than one solution corresponding to different initial value problem we can use the same **DEplot** command specifying the list of initial conditions:

```
> DEplot(deq1,y(x),x=-
2..2,[[y(0)=1,D(y)(0)=0],[y(0)=1,D(y)(0)=1],[y(0)=0,D(y)(0)=0]],y=-
10..10,linecolor=red);
```



The general second order linear DE, y " + P(x) y ' + Q(x) y = f(x):

Note: Maple is unable to solve most second-order DE's explicitly. For information on numerically solving DE's, see Numerical Solutions with dsolve.

Consider the differential equation y" + x y'(x) + y(x) = sin(x). Try to use the **dsolve** command.

```
> deq2:=diff(y(x),x$2)+x*diff(y(x),x)+y(x)=sin(x);
```

$$deq2 := \left(\frac{d^2}{dx^2} y(x)\right) + x\left(\frac{d}{dx} y(x)\right) + y(x) = \sin(x)$$

```
> dsolve(deq2,y(x));
```

$$y(x) = e^{\left(-\frac{x^2}{2}\right)} \_C1 \operatorname{erf}\left(\frac{1}{2} I \sqrt{2}\, x\right) + e^{\left(-\frac{x^2}{2}\right)} \_C2 + \frac{1}{4} I \sqrt{2} \sqrt{\pi}\, e^{(1/2)} \left(\operatorname{erf}\left(\frac{1}{2} I \sqrt{2}\, x - \frac{\sqrt{2}}{2}\right) + \operatorname{erf}\left(\frac{1}{2} I \sqrt{2}\, x + \frac{\sqrt{2}}{2}\right)\right) e^{\left(-\frac{x^2}{2}\right)}$$

```
> in_cond2:=y(0)=1,D(y)(0)=1;
```

$$in\_cond2 := y(0) = 1, D(y)(0) = 1$$

```
> dsolve({deq2,in_cond2},y(x));
```

$$y(x) = -e^{\left(-\frac{x^2}{2}\right)} \sqrt{\pi} \sqrt{2} \operatorname{erf}\left(\frac{1}{2} I \sqrt{2}\, x\right) I + e^{\left(-\frac{x^2}{2}\right)} + \frac{1}{4} I \sqrt{2} \sqrt{\pi}\, e^{(1/2)} \left(\operatorname{erf}\left(\frac{1}{2} I \sqrt{2}\, x - \frac{\sqrt{2}}{2}\right) + \operatorname{erf}\left(\frac{1}{2} I \sqrt{2}\, x + \frac{\sqrt{2}}{2}\right)\right) e^{\left(-\frac{x^2}{2}\right)}$$

Maple expresses the solution in terms of a modified Bessel function I and the error function **erf**.
We can obtain the numerical solution using in the dsolve command the option '**type=numeric**' and the **odeplot** comand to draw the corresponding graph.

```
> n_sol:=dsolve({deq2,in_cond2},y(x),type=numeric):
> odeplot(n_sol);
```




```
>
```